FYS4310. Problem 300-2. Suggested solution

Problem: construct semilog plot using table 3.2 and 3 contributions to diffusivity of phosphorous as a function of temperature from 700 ..1100 C
assume C of phosphorous is 1e19 cm^-3

```
> restart;
```

Symbol definitions
*D*o = Diffusivity of P by neutral vacancy in intrinsic Si at infinite temperature in cm^2/s
*E*a = activation energy for above in eV
*D*om = (*D* zero minus) Diffusivity of P by negative vacancy in i-Si at inf. temperature
*E*am = activation energy for above
*D*odm =(*D* zero double minus) Diffusivity of P by double negative vacancy diff in i-Si at inf. temperature
*E*adm = activation energy for above

Bolzmanns constant
T absolute temp i K
n = electron concentration in cm^-3
*n*i = intrinsic carrier concentration
*n*io = constant giving the intrinsic concentration
*E*g = Bangap of Si in eV
*E*go=Bangap at T=0
alpha, beta = constants
deltaEg = decrease in Eg at high doping
*N*D=Donor concentration

```
> #assume(T>0,n>0,ni>0,ND>0,Ea>0,Eadm>0,Eam>0,Do>0,Dom>0,Dodm>0,k>0,
  nio>0,alpha>0,beta>0);
```

The Diffusivity of P in Si can be written as

$$> eq1 := \mathrm{D} = Do\, \mathrm{e}^{-\frac{Ea}{k\,T}} + \frac{Dom\, \mathrm{e}^{-\frac{Eam}{k\,T}}\, n}{ni} + Dodm\, \mathrm{e}^{-\frac{Eadm}{k\,T}} \left(\frac{n}{ni}\right)^2$$

$$eq1 := \mathrm{D} = Do\, \mathrm{e}^{-\frac{Ea}{k\,T}} + \frac{Dom\, \mathrm{e}^{-\frac{Eam}{k\,T}}\, n}{ni} + \frac{Dodm\, \mathrm{e}^{-\frac{Eadm}{k\,T}}\, n^2}{ni^2}$$

The intrinsic carrierconcentration is given by (see 3.8 in textbook)

$$> eq2 := ni = nio\, T^{\frac{3}{2}}\, \mathrm{e}^{-\frac{Eg}{2\,k\,T}}$$

$$eq2 := ni = nio\, T^{3/2}\, \mathrm{e}^{-\frac{1}{2}\,\frac{Eg}{k\,T}}$$

The bandgap is given by (exept for highdoping effects) is given by (text book 3.9)

$$> eq3 := Eg = Ego - \frac{\alpha\, T^2}{\beta + T}$$

$$eq3 := Eg = Ego - \frac{\alpha T^2}{\beta + T}$$

We discussed in the class wether to additionally include the bandgap narrowing due to high concentration. Some of that discussion can be found in another worksheet.
We decided to not include heavy doping effects on the bandgap because
a) tgf did not believe the formulaes we have available yeld numerically reliable results.
b) tgf thinks the underlying theoretical foundation is lacking and possibly quoted incorrect
c) other corrections should also be applied for example the Fermi integral would theoretically describe the electron concentration than the Bolzman approximation
d) The high doping bandgap narrowing was not used when the diffusivities given in the table of the book was extracted from experimental data (again from concentration profiles)

So we do not here use equation (.3.10) and learned that 'life is not simple'

We have given the Phosphorous concentration, ND, we need to find the electrom concentration, n.
If we have no segregation or complex formation between point defects the relationship between them is given by charge neutrality yielding

```
> eq4 := n = ND + ni^2/n
```

$$eq4 := n = ND + \frac{ni^2}{n}$$

Which came from requirement of charge neutrality: ND+ - NA- + p - n = 0 and assume complite ionization of donors b setting ND+ = ND.
We solve eq4 for n

```
> eq5:=solve(eq4,n);
```

$$eq5 := \frac{1}{2} ND + \frac{1}{2}\sqrt{ND^2 + 4\,ni^2},\ \frac{1}{2} ND - \frac{1}{2}\sqrt{ND^2 + 4\,ni^2}$$

We definitely have to pick the solution with positive sign

```
> if (is(op(0,eq5[1])=`+`)) then eq7:=n=eq5[1] else eq7:=n=eq5[2];
  fi;
```

$$eq7 := n = \frac{1}{2} ND + \frac{1}{2}\sqrt{ND^2 + 4\,ni^2}$$

The numerical values of parameters in the previous equations are (as read from table 3.2, and on page 42 -43) in units based on cm , eV, s and K

```
> constants:={k=8.617065e-5, alpha=0.000473,Ego=1.17,beta=636,nio=
  7.3e15, Do=3.9,Ea=3.66,Dom=4.4,Eam=4,Dodm=44,Eadm=4.37,ND=1e19};
```

$$constants := \{Dodm = 44, Dom = 4.4, Do = 3.9, Eadm = 4.37, Eam = 4, Ea = 3.66, Ego = 1.17, ND = 1.\ 10^{19}, \alpha = 0.000473, \beta = 636, k = 0.00008617065, nio = 7.3\ 10^{15}\}$$

Let us just estimate if we need to use equation eq7 or can use n=ND.
We calculate ni at 1000K and 1400K at low doping
We thus use the following equation

```
> eq8:=subs(eq3,eq2);
```

$$eq8 := ni = nio\ T^{3/2}\ e^{-\frac{1}{2}\ \frac{Ego - \frac{\alpha T^2}{\beta + T}}{k T}}$$

We calculate at a)T=1000K, b)T=1400K c)T=1273K (=1000 C)

We just put in the values of the symbols, calc at T=1000 and 1400 , and 1000C=1273 respectively

```
> eq9a:=evalf(subs(constants union {T=1000,ND=1e19},eq8));
  eq9b:=evalf(subs(constants union {T=1400,ND=1e19},eq8));
  eq9c:=evalf(subs(constants union {T=1273,ND=1e19},eq8));
```

$$eq9a := ni = 1.391660694\ 10^{18}$$

$$eq9b := ni = 1.977654144\ 10^{19}$$

$$eq9c := ni = 9.984573047\ 10^{18}$$

So we can culculate n

```
> eq10a:=subs(constants union {eq9a},eq7);
  eq10b:=subs(constants union {eq9b},eq7);
  eq10c:=subs(constants union {eq9c},eq7);
```

$$eq10a := n = 1.019005968\ 10^{19}$$

$$eq10b := n = 2.539881348\ 10^{19}$$

$$eq10c := n = 1.616654373\ 10^{19}$$

We have presented the equations we use, from this we can make the diffusivity an explicit function of temperature
We put eq8 into eq7 to get n and put n and ni (from eq8) into eq1 ,
The line below is doing exactly that

```
> DofT:=T->subs(constants,subs(subs(eq8,eq7),subs(eq8,eq1))):
```

We may also make each of the diffusion terms in eq1 an explicit function of temperature
In the following lines first for neutral vacancy contribution, negative vacancy and double negative vacancy.

```
> Do_ofT:=T->subs(constants,subs(subs(eq8,eq7),subs(eq8,Do*exp(-
  Ea/k/T)))):
> Dm_ofT:=T->subs(constants,subs(subs(eq8,eq7),subs(eq8,Dom*exp(-
  Eam/k/T)*n/ni))):

> Ddm_ofT:=T->subs(constants,subs(subs(eq8,eq7),subs(eq8,Dodm*exp(-
  Eadm/k/T)*(n/ni)^2))):
```

We will calculate the diffusivity for a range of temperatures and plot
We decide to plot log(D) versus 1/T, thus we will pick values of (1/T) that are evenly spaced
We will do this in a procedure

```
> with(plots):
> Makeplots:=proc()
  local j, invT, delinvT, TK, D , LogDarray,
        vD,vDo,vDm,vDdm,LogDoarray,LogDmarray,LogDdmarray,invTarray,
        dplot,doplot,dmplot,ddmplot:
  invT:=1.0/1400:## invT=1/T, we start at the highest temparature
  delinvT:=evalf((1/1000 -1/1400)/20.0):# we calculate the step size
  in 1/T
  for j from 1 to 20 do
   TK:=1/invT:
   vDo:=Do_ofT(TK);
   vDm:=evalf(eval(Dm_ofT(T),T=TK)):
   vDdm:=evalf(eval(Ddm_ofT(T),T=TK)):
   vD:=vDo+vDm+vDdm:
   LogDarray[j]:=evalf(log10(vD)):
   LogDoarray[j]:=evalf(log10(vDo)):
   LogDmarray[j]:=evalf(log10(vDm)):
   LogDdmarray[j]:=evalf(log10(vDdm)):
   invTarray[j]:=invT:
   invT:=invT+delinvT:
  od:
  dplot:=listplot([seq([invTarray[i],LogDarray[i]],i=1..20)],color=
  red):
  doplot:=listplot([seq([invTarray[i],LogDoarray[i]],i=1..20)],color=
  cyan):
  dmplot:=listplot([seq([invTarray[i],LogDmarray[i]],i=1..20)],color=
  green):
  ddmplot:=listplot([seq([invTarray[i],LogDdmarray[i]],i=1..20)],
  color=blue):
  display({dplot,doplot,dmplot,ddmplot});
  end:

> Makeplots();
```



The above plot shows as
red = Total diffusivity
Cyan = the contribution from neutral vacancies
Green = the contribution from negative vacancies
Blue = the contribution from double negative vacancies
Axes horizontal 1/T in (K^-1) vertical log10 to D

We should compare the above figure with Data given in other books,
But textbooks only gives Do and Ea values and figures and we dont know what they use for n and ni to extract the parameters.

> The End